
EventStreamex
    

Package to add event streaming in your Elixir application, using PostgreSQL WAL.
Works very well with Phoenix LiveView to add realtime to your app.
This package uses the Write-Ahead-Log from PostgreSQL to listen for changes in database tables and then, handle them in your application in 2 ways:
	To add realtime support in your views
	To update other entities in realtime (as soon as an entity it depends on is updated)

Full documentation can be found here.

  
    
  
  Usage



  
    
  
  Installation


The package can be installed by adding event_streamex to your list of dependencies in mix.exs:
def deps do
  [
    {:event_streamex, "~> 1.3.0"}
  ]
end

  
    
  
  Configuration


Here are the configuration parameters for EventStreamex with their default value:
config :event_streamex,
  app_name: nil,
  database: "postgres",
  durable_slot: true,
  error_logger_adapter: {EventStreamex.Operators.Logger.LoggerAdapter, []},
  hostname: "localhost",
  operator_queue_backoff_multiplicator: 2,
  operator_queue_max_restart_time: 10000,
  operator_queue_max_retries: 5,
  operator_queue_min_restart_time: 500,
  password: "postgres",
  port: "5432",
  process_status_entity_field: :updated_at,
  process_status_storage_adapter: {EventStreamex.Operators.ProcessStatus.DbAdapter, []},
  publication: "events",
  pubsub: [adapter: Phoenix.PubSub, name: nil],
  queue_storage_adapter: {EventStreamex.Operators.Queue.DbAdapter, []},
  slot_name: "postgres_slot",
  url: "",
  username: "postgres"
	app_name: Your application name
	database: Name of the database to connect to
	durable_slot: Is the replication slot for the WAL temporary or durable? If you set it to false, then, you will loose all WAL events sent while your application was not running.
	error_logger_adapter: The logger module used with the start_link parameter to send it
	hostname: Your PostgreSQL hostname
	operator_queue_backoff_multiplicator: Multiplicator used to compute the new wait time after an operator crash (See EventStreamex.Operators.Logger.ErrorLoggerAdapter)
	operator_queue_max_restart_time: Maximum time to wait in milliseconds before the restart of a crashed operator (See EventStreamex.Operators.Logger.ErrorLoggerAdapter)
	operator_queue_max_retries: Maximum number of restarts of a crashed operator (See EventStreamex.Operators.Logger.ErrorLoggerAdapter)
	operator_queue_min_restart_time: Time to wait in milliseconds before the first restart of a crashed operator (See EventStreamex.Operators.Logger.ErrorLoggerAdapter)
	password: Password used to connect to the database
	port: Port of the database
	process_status_entity_field: The field name used is your entities to represent the last updated time. It must be set also when the entity is created. And it should have a precision of microseconds (use the utc_datetime_usec type with Ecto). This field is used to update the EventStreamex.Operators.ProcessStatus registry.
	process_status_storage_adapter: Module used for process status storage and the parameter sent to the start_link function
	publication: Name of the publication used by the WAL (it is created automatically if it does not already exist)
	pubsub: Adapter to use for pubsub and its name
	queue_storage_adapter: Module used for queue storage and the parameter sent to the start_link function
	slot_name: Name of the replication slot for the WAL. It is automatically created if it does not exist.
	url: URL to connect to the database. If it is not set, the other parameters will be used instead
	username: Username to connect to the database

For the package to work correctly you will have to specify, at least:
	The database connection parameters
	The :app_name
	The pubsub :name


  
    
  
  Mark the entity for listening


To mark an entity for listening, you will have to do 2 things:
	use the EventStreamex.Events module
	and set the tables for replication

Use the EventStreamex.Events module
defmodule MyApp.Blog.Post do
  use MyApp.Schema

  use EventStreamex.Events, name: :my_app, schema: "posts"

  import Ecto.Changeset

  alias MyApp.Blog.PostWithCommentsCount

  schema "posts" do
    field :title, :string
    field :body, :string

    timestamps(type: :utc_datetime)
  end

  @doc false
  def changeset(post, attrs) do
    post
    |> cast(attrs, [:id, :title, :body])
    |> validate_required([:title, :body])
  end
end
Set the tables for replication
For each table of your application, you will have to execute this SQL query (ideally in the migrations);
ALTER TABLE ONLY <table_name> REPLICA IDENTITY FULL;
Of course, you will replace <table_name>by the name of the table.
Any marked entity will tell PostreSQL's WAL to emit events as soon as it is inserted, updated, deleted.

  
    
  
  Listen to events in live views


Now that our posts entity is subscribed to, in the WAL, we can listen to its changes in a live view:
defmodule MyApp.PostLive.Index do
  use MyApp, :live_view

  use EventStreamex.EventListener,
    schema: "post_with_comments_counts",
    subscriptions: [:unscoped]

  alias MyApp.Blog
  alias MyApp.Blog.Post

  @impl true
  def mount(params, session, socket) do
    {:ok, socket} = super(params, session, socket)

    {:ok, stream(socket, :posts, Blog.list_posts())}
  end

  @impl true
  def handle_params(params, url, socket) do
    {_res, socket} = super(params, url, socket)

    {:noreply, socket}
  end

  @impl true
  def handle_info({:on_insert, [], "post_with_comments_counts", post}, socket) do
    {:noreply,
     socket
     |> stream_insert(:posts, post.new_record)
  end

  @impl true
  def handle_info({:on_update, [], "post_with_comments_counts", post}, socket) do
    {:noreply,
     socket
     |> stream_insert(:posts, post.new_record)
  end

  @impl true
  def handle_info({:on_delete, [], "post_with_comments_counts", post}, socket) do
    {:noreply,
     socket
     |> stream_delete(:posts, post.old_record)
  end
end
Here, we listen to posts events in our live view.
The handle_info/2 callbacks are called with the kind of event received (:on_insert, :on_update, :on_delete) and the event.
You can, then do whatever you need to update your view.

  
    
  
  Events structure


An event looks like this:
%WalEx.Event{
  name: atom(),
  type: :insert | :update | :delete,
  source: %WalEx.Event.Source{
    name: String.t(),
    version: String.t(),
    db: String.t(),
    schema: String.t(),
    table: String.t(),
    columns: map()
  },
  new_record: map() | nil,
  old_record: map() | nil,
  changes: map() | nil,
  timestamp: DateTime.t(),
  lsn: {integer(), integer()}
}
I am using the WalEx package internally
	name: The name of the entity (ie: :comments for a table named comments)
	type: The type of event between insert, update and delete
	source: Information about the event:	name: "WalEx"
	version: Current version of WalEx
	db: The name of the database
	schema: Mostly "public"
	table: The name of the table (ie: "comments")
	columns: A map of fields with their type (ie: %{"id": "integer", "message": "varchar"})


	new_record: The entity itself for insert and update events. nil for delete events.
	old_record: The entity itself for delete events. nil for insert and update events.
	changes: A map with the changes in the entity in update events, nil otherwise (see below)
	timestamp: The timstamp of the event in DateTime type
	lsn: A tuple containing information about the publication cursor

changes
When you receive an update event, you will also have the changes field set to a map containing
the changes the entity received since the update.
This map contains the changed fields as keys, and a map describing the change as value.
This "change" map contains 2 fields:
	old_value: The value before the update
	new_value: The value after the update

For instance, let's say you have a comments entity with 4 fields: id, message, rating, timestamp.
You have a comment with these values:
%Comment{
  id: "dd4bc2ba-c7cc-4a05-a1c7-9f26cd9ab79f",
  message: "This is my first comment",
  rating: 4,
  timestamp: "2024-07-23T12:00:00Z"
}
Now, the comment is update this these new values:
%Comment{
  id: "dd4bc2ba-c7cc-4a05-a1c7-9f26cd9ab79f",
  message: "This is (not) my first comment (anymore)",
  rating: 5,
  timestamp: "2024-07-23T12:00:00Z"
}
The event structure will look like this:
%WalEx.Event{
  name: :comments,
  type: :update,
  source: %WalEx.Event.Source{
    name: "WalEx",
    version: "4.4.0",
    db: "postgresql",
    schema: "public",
    table: "comments",
    columns: %{
      id: "uuid",
      message: "varchar",
      rating: "integer",
      timestamp: "datetime"
    }
  },
  new_record: %Comment{
    id: "dd4bc2ba-c7cc-4a05-a1c7-9f26cd9ab79f",
    message: "This is (not) my first comment (anymore)",
    rating: 5,
    timestamp: "2024-07-23T12:00:00Z"
  },
  old_record: nil,
  changes: %{
    message: %{
      old_value: "This is my first comment",
      new_value: "This is (not) my first comment (anymore)"
    },
    rating: %{
      old_value: 4,
      new_value: 5
    }
  },
  timestamp: "2024-08-25T13:13:30Z",
  lsn: {0, 0}
}

  
    
  
  Update other entities based on events


Now, let's assume that we also have a comments entity linked to a post,
and that we want to display our posts with the number of comments they have.
In event streaming we will create another entity (let's say post_with_comments_count), and update it when we receive events from posts and comments.
This entity will look like this:
	id: It's id (same one as the post id)
	title: Same one as the post title
	body: Same one as the post body
	comments_count: The number of comments of the post

defmodule MyApp.Operators.PostWithCommentsCountOperator do
  use EventStreamex.Operators.Operator,
    schema: "post_with_comments_counts",
    repo: MyApp.Repo

  on_event(
    all_events(),
    entities(["comments", "posts"]),
    query(
      """
      MERGE INTO post_with_comments_counts dest
      USING (
          SELECT
              p.id AS id,
              p.title AS title,
              p.body AS body,
              p.inserted_at AS inserted_at,
              p.updated_at AS updated_at,
              (SELECT COUNT(c.id) FROM comments c WHERE c.post_id = p.id) AS comments_count
          FROM posts p
          WHERE p.id = $1
      ) src ON src.id = dest.id
      WHEN NOT MATCHED THEN
        INSERT (id, title, body, inserted_at, updated_at, comments_count)
        VALUES (src.id, src.title, src.body, src.inserted_at, src.updated_at, src.comments_count)
      WHEN MATCHED THEN
        UPDATE SET
            title = src.title,
            body = src.body,
            inserted_at = src.inserted_at,
            updated_at = src.updated_at,
            comments_count = src.comments_count;
      """,
      id([{"comments", :post_id}, {"posts", :id}]),
      delete_with(["posts"])
    )
  )
end
Using an operator will allow us to insert/update/delete an entity based on events.
Here, we listen to posts and comments and, for each event, we execute the MERGE query.
This query will create the post_with_comments_count when a post is created, and update it when the post is updated.
It will also update the comments_count each time there is a new comment,
or a comment has been deleted.
The $1 parameter in the query will be replaced by the field in the listened entities that corresponds to the id of the post_with_comments_count.
In this example, it's the post ID. Which is found in the post_id field in comments.
That comes from this line: id([{"comments", :post_id}, {"posts", :id}]).
delete_with(["posts"]) will tell the operator that a post deletion must also delete the post_with_comments_count.
If you can more complex scenarios, you can also use a function instead of query/3:
on_event(
  all_events(),
  entities(["comments", "posts"]),
  fn event ->
    {:ok, res} =
      event
      |> MyQuery.do_some_queries()
      |> MyQuery.then_another_query()
    {:ok, res}
  end
)

  
    
  
  Why using event streaming?


Although using event streaming with EventStreamex looks easy, why bother using event streaming in the first place?

  
    
  
  The issue without event streaming


Let's say you have a posts entity which contains:
	title: a title
	body: a body

You also have a comments entity which contains:
	post_id: The ID of the post it is attached to
	comment: The comment content

When you show a list of your posts you might want to show the number
of comments it contains.
To do so you will be tempted to make a SQL query with a subquery that will
count the comments of your posts.
This is pretty trivial:
SELECT
  p.id AS id,
  p.title AS title,
  p.body AS body,
  (SELECT COUNT(*) FROM comments c WHERE c.post_id = p.id) AS comments_count
FROM posts p;
But it has several flaws.
Let's see them in more details.
Your queries will become slower and slower over time
Because subqueries are not free. They take more time to execute that a simple query with no join and subquery.
So, the more posts and comments you have, the slower the query will be.
You will end up adding complexity in your queries
As your app evolves, you will probably create new entities related to your posts and comments.
Thus, you will add more and more subqueries and joins to your queries, making them slower and harder to maintain...
Realtime will become a mess to maintain
If you want to handle realtime events like updating the number of comments when a new one is created it will be a mess to maintain overtime because you will end up subscribing to a lot of events as you link your posts to other entities.
And you will probably forget to subscribe to some of the events making your app inconsistant (some features will be in realtime and other won't be).
Almost impossible to maintain recursivity
If your comments can have subcomments it will be almost impossible
to maintain the comments count because of the recursivity it implies.
Depending on entities not directly related to the base entity
Let's say you create another entity named ratings which saves each rating given to a specific comment.
It would look like this:
	comment_id: The ID of the comment related to this rating
	rating: The given rating

Now imagine you want to display, for every comment, the average rating and the total count of ratings for this comment.
Here again, you could solve this with some subqueries.
But what if you want to show the average rating of all comments in the post
so that users can see that a post has received a lot of useful comments.
I don't even want to think about the query that would require...

  
    
  
  Benefits from using event streaming


Let's come back to our posts and comments.
Now let's imagine that instead of our subquery to count the comments we use
an event streaming architecture and create another entity whose goal is
specifically to show the posts with their comments count.
This post_with_comments_counts entity will look like that:
	title: a title
	body: a body
	comments_count: the number of comments of the post

And this entity will listen to new comments to update its comments count:
when a new comment is created it will increase the number of comments of the related post by 1.
With EventStreamex it would look like this:
defmodule MyApp.Operators.PostWithCommentsCountOperator do
  use EventStreamex.Operators.Operator,
    schema: "post_with_comments_counts",
    repo: MyApp.Repo

  on_event(
    all_events(),
    entities(["comments", "posts"]),
    query(
      "MERGE INTO post_with_comments_counts dest
      USING (
          SELECT
              p.id AS id,
              p.title AS title,
              p.body AS body,
              p.inserted_at AS inserted_at,
              p.updated_at AS updated_at,
              (SELECT COUNT(c.id) FROM comments c WHERE c.post_id = p.id) AS comments_count
          FROM posts p
          WHERE p.id = $1
      ) src ON src.id = dest.id
      WHEN NOT MATCHED THEN
        INSERT (id, title, body, inserted_at, updated_at, comments_count)
        VALUES (src.id, src.title, src.body, src.inserted_at, src.updated_at, src.comments_count)
      WHEN MATCHED THEN
        UPDATE SET
            title = src.title,
            body = src.body,
            inserted_at = src.inserted_at,
            updated_at = src.updated_at,
            comments_count = src.comments_count;",
      id([{"comments", :post_id}, {"posts", :id}]),
      delete_with(["posts"])
    )
  )
end
Here, we use the EventStreamex.Operators.Operator module to listen for
posts and comments in all kinds of events (insert, update, delete).
And each time a post or a comment is inserted, updated, deleted, we perform this MERGE query to create/update our post_with_comments_counts.
The MERGE query will insert a new post if the post does not exist and update it if it already exists.
id/1 is used to map the fields of the listened entities to the id of the
post_with_comments_counts entity (id for a post and post_id for a comment).
Finally, delete_with/1 tells the operator which entities deletion should
also delete the related post_with_comments_counts entity.
As you can see, this MERGE query also uses a subquery, so why is it better than our first implementation?
This is a naive implementation
The query/3 macro is provided for simplicity.
But you can also replace it by a function to do things as you think it's best (optimize the queries according to the type of event and entity received).
This function will receive the event, allowing you to do whatever you want with it.
The query is executed in the background
In our first example, the query was executed everytime someone wants to display the list of posts.
In our event streaming example, the query is executed in the background each time there is an event with a post or a comment.
To display your list of posts you will just have to do:
SELECT * FROM post_with_comments_counts;
Which is simpler and faster than our former query.
Realtime is easy
You can use the module EventStreamex.EventListener to easily add realtime
in your live view applications.
Because we are using PostgreSQL WAL events, everytime something happens on an entity (post_with_comments_counts as well), an event is sent.
This event will be processed by operators listening to it (like our operator here does for posts and comments), and it will also be sent to a pubsub processor.
The EventStreamex.EventListener, now just has to listen to this pubsub processor to notify your live view that the entity your are showing as been updated.
Everything, now, can be realtime, without beeing hard to maintain.
A better pattern to model your data
With this approach you keep your entities consistent and simple.
And you create specialized entities for what you want to show in your views.
A post will stay a post with its title and body fields and a comment will stay a comment with a comment field.
post_with_comments_count is just a specialized entity used for your view.
This keeps your data and APIs consistent and simple with explicit fields and names.
For instance, let's say you have a users entity with a login and a password field.
But later on, you realize that you need to store some information about your users, like its name, its address, and so on.
You will be tempted to update your users entity to add these new fields:
	login
	password
	address
	name

But still, in your application, when comes the moment to create your user, you only need the login and password.
So what do you do? You set the other fields as optionals in your API and set them a default value (empty string or null).
And in the view that allows a user to specify more information about him (name and address), you will either:
	Use the same API endpoint as for the users creation
	Or create a new dedicated API endpoint

But both approachs are bad practices!
In the first approach that means that your API expects 2 mandatory fields: login and password, and 2 optional fields: name and address.
But here, there is no sense in sending the login and password.
And maybe you would like the name and address fields to be mandatory.
But because your API set these fields as optionals, you will have to check for them from the client side.
Which is a bad practice.
In the second approach, that means that you will create an API endpoint for the user profile.
But this API, will finally update the users entity.
You end up with hard to understand APIs because their name does not match
to the entity being modified.
A better approach would be to have a users entity with:
	login
	password

And its own API endpoint /users with only these two fields.
And you would have a user_profiles entity with:
	user_id
	name
	address

And its own API endpoint /users/{user_id}/user_profiles with only the name and address fields.
And what if you need to display all of these information in your view?
Well, you use an operator like we did for the post_with_comments_counts.
Named, for instance: user_with_profiles, and listening to users and user_profiles events.
This way, you keep your data and API consistent.

  
    
  
  How to handle asynchronicity of entity creation


Event streaming is nice, but it makes create/update/delete operations asynchronous for derived entities.
Which can be a nightmare to handle correctly.
Let's say you have a LiveView application showing a list of post_with_comments_counts, which is a derived entity from the posts entity, containing, in addition to posts fields, a comments_count field.
From this view, we can also create a new post.
But we don't create a post_with_comments_counts directly.
Instead, we must create a posts entity.
Which, in turn, will trigger the creation of the post_with_comments_counts using event streaming.
What we want to do is to redirect to the post detail page when the post is
created.
But we have a problem!
Because the creation of the posts entity is synchronous, we can redirect
as soon as it is created.
But that does not mean the post_with_comments_counts entity will be created already... while it's this entity we need to show.
To handle this scenario and be able to display the post_with_comments_counts entity as soon as it is created we have 3 solutions:

  
    
  
  Wait for the entity to be created before redirecting


To ensure you redirect to an entity that exists, you can actualy wait for it
to be created.
In order to do so, you will have to listen to this entity to know when it's been created.
You can use the subscribe_entity/4 function for that matter.
defmodule MyApp.PostLive.Index do
  use MyApp, :live_view

  use EventStreamex.EventListener,
    schema: "post_with_comments_counts",
    subscriptions: [:unscoped]

  alias MyApp.Blog
  alias MyApp.Blog.Post

  @impl true
  def mount(params, session, socket) do
    {:ok, socket} = super(params, session, socket)

    {:ok, stream(socket, :posts, Blog.list_posts())}
  end

  @impl true
  def handle_params(params, url, socket) do
    {_res, socket} = super(params, url, socket)

    {:noreply, apply_action(socket, socket.assigns.live_action, params)}
  end

  defp apply_action(socket, :new, _params) do
    socket
    |> assign(:page_title, "New Post")
    |> assign(:post, %Post{})
  end

  defp apply_action(socket, :index, _params) do
    socket
    |> assign(:page_title, "Listing Posts")
    |> assign(:post, nil)
  end

  @impl true
  def handle_info({MyApp.PostLive.FormComponent, {:saved, post}}, socket) do
    {:noreply,
     socket
     |> subscribe_entity("post_with_comments_counts", :direct, %{"id" => post.id})}
  end

  @impl true
  def handle_info({:on_insert, :direct, "post_with_comments_counts", post}, socket) do
    {:noreply,
     socket
     |> unsubscribe_entity("post_with_comments_counts", :direct)
     |> push_navigate(to: "/posts/#{post.new_record.id}")}
  end

  @impl true
  def handle_info({_, :direct, "post_with_comments_counts", _post}, socket) do
    {:noreply, socket}
  end

  @impl true
  def handle_info({:on_insert, [], "post_with_comments_counts", post}, socket) do
    {:noreply,
     socket
     |> stream_insert(:posts, post.new_record)
  end

  @impl true
  def handle_info({:on_update, [], "post_with_comments_counts", post}, socket) do
    {:noreply,
     socket
     |> stream_insert(:posts, post.new_record)
  end

  @impl true
  def handle_info({:on_delete, [], "post_with_comments_counts", post}, socket) do
    {:noreply,
     socket
     |> stream_delete(:posts, post.old_record)
  end
end
Here, our live view listens for :unscoped post_with_comments_counts entity for the index view in realtime (nothing new).
This code is more interesting:
@impl true
def handle_info({MyApp.PostLive.FormComponent, {:saved, post}}, socket) do
  {:noreply,
    socket
    |> subscribe_entity("post_with_comments_counts", :direct, %{"id" => post.id})}
end
This function gets called when a posts has been created.
And inside it, instead of redirecting immediately, we call the subscribe_entity/4 function to listen for post_with_comments_counts with a specific ID. The ID of the newly created post.
That means that as soon as this post_with_comments_counts entity will be created, we will be notified:
@impl true
def handle_info({:on_insert, :direct, "post_with_comments_counts", post}, socket) do
  {:noreply,
    socket
    |> unsubscribe_entity("post_with_comments_counts", :direct)
    |> push_navigate(to: "/posts/#{post.new_record.id}")}
end
When the entity is created, we unsubscribe from the channel (this is optional is it will automatically unsubscribe as soon as the live view terminates), and we redirect to the detail page.
And we know, at this moment, the post_with_comments_counts entity will exist.

  
    
  
  Handle entity not found


Another solution is to redirect immediately without waiting for the post_with_comments_counts to be created, and by handling the case where the entity is not found yet.
defmodule MyApp.PostLive.Show do
  use MyApp, :live_view

  use EventStreamex.EventListener,
    schema: "post_with_comments_counts",
    subscriptions: [:direct]

  alias MyApp.Blog

  @impl true
  def mount(params, session, socket) do
    super(params, session, socket)
  end

  @impl true
  def handle_params(%{"id" => id} = params, url, socket) do
    {_res, socket} = super(params, url, socket)

    post = Blog.get_post(id)

    {:noreply,
     socket
     |> assign(:id, id)
     |> assign(:post, post)
  end

  @impl true
  def handle_info({:on_insert, :direct, "post_with_comments_counts", post}, socket) do
    {:noreply,
     socket
     |> assign(:post, post.new_record)}
  end

  @impl true
  def handle_info({:on_update, :direct, "post_with_comments_counts", post}, socket) do
    {:noreply,
     socket
     |> assign(:post, post.new_record)}
  end

  @impl true
  def handle_info({:on_delete, :direct, "post_with_comments_counts", _post}, socket) do
    {:noreply, socket |> put_flash(:warning, "Ce post vient d'être supprimé")}
  end
end
Here we fecth the entity (but without triggering an exception if it does not exist), and we replace it if we receive the :on_insert event for that entity.
You will just have to ensure your view shows a loader or something until the entity is created.

  
    
  
  Handling async with APIs


We saw 2 solutions we could use with a live view.
But what if you develop an mobile app and use an API?
If you have an API endpoint to create a new posts (POST /posts), and as soon as your appliation calls it, it redirects to another view requesting the detail of the created post_with_comments_counts (GET /post_with_comments_counts), then, you may receive a 404 error because of the async creation of the post_with_comments_counts entity.
In order to avoid that, you could implement a mecanism that would return, along with the posts newly created, the post creation timestamp.
Then, in the GET /post_with_comments_counts endpoint, you would send along the timestamp you got from the POST /posts endpoint.
And in your controller you would check if the entity has been processed before returning the post_with_comments_counts requested.
You can use EventStreamex.Operators.ProcessStatus.processed?/2 in order to do so.
For instance, when you create a posts, you receive this header from the API:
X-Event-Streamex-Entity-Timestamp: "posts/1422057007123000"
def create(conn, %{"post" => post_params}) do
  with {:ok, %Post{} = post} <- Blog.create_post(post_params) do
    conn
    |> put_status(:created)
    |> put_resp_header("location", ~p"/api/posts/#{post}")
    |> put_resp_header(
      "x-event-streamex-entity-timestamp",
      "posts/#{EventStreamex.Operators.EntityProcessed.to_unix_timestamp!(post.updated_at)}"
    )
    |> render(:show, post: post)
  end
end
(You can use the :process_status_entity_field config parameter to set the field to use in your entities to know the last updated time. You should return the field's value in your API, and use this value in the processed?/2 function)
You will send this header when you call the GET /post_with_comments_counts endpoint.
And the controller will immediately check if the entity has been processed:
EventStreamex.Operators.ProcessStatus.processed?("posts", 1422057007123000)
If the entity has not been processed yet, then, you know the post_with_comments_counts entity has not been created yet (it works the same for updated entities).
So you can immediately return a 425 response for instance, informing the client to retry later.
If the entity has been processed, then you fetch it and return it.
If you get a 404 error, this time, that means the entity did not exist.
You can do that in a plug in your api scope:
defmodule MyApp.Plugs.EventStreamexEntityTimestamp do
  import Plug.Conn

  @header_name "x-event-streamex-entity-timestamp"

  def init(default), do: default

  def call(%Plug.Conn{} = conn, _default) do
    case get_req_header(conn, @header_name) do
      [] ->
        conn

      [entity_timestamp] ->
        case decode_entity_timestamp(entity_timestamp) do
          {:ok, entity, timestamp} ->
            unless EventStreamex.Operators.ProcessStatus.processed?(entity, timestamp) do
              conn
              |> put_status(425)
              |> resp(425, "Entity not processed yet")
              |> halt()
            else
              conn
            end

          _ ->
            conn
        end
    end
  end

  def decode_entity_timestamp(entity_timestamp) do
    case entity_timestamp |> String.split("/") do
      [entity | [timestamp_str | []]] ->
        case Integer.parse(timestamp_str, 10) do
          {timestamp, _} ->
            {:ok, entity, timestamp}

          :error ->
            {:error, :not_integer}
        end

      _ ->
        {:error, :bad_format}
    end
  end
end
Of course you can implement it the way you want, but you've got the idea.

  
    
  
  Tests


To test the package you can run:
mix test
Some tests will need access to a PostgreSQL database.
Here are the credentials used by the tests:
	Database: postgres
	Hostname: locahost
	Password: ""
	Port: 5432
	Username: postgres

The event streaming system supervisor.
This supervisor is the one started by EventStreamex.Orchestrator when
the current node is the master node.
This supervisor is responsible for starting the whole system.

Listens for new database WAL events in a Phoenix.LiveView.
As soon as something happens in the database (insert, delete, update),
a WAL event is sent and dispatched to operators that listen to it, and to
live views that uses the EventStreamex.EventListener module.
Let's say you have an entity named comments and you have a live view that
shows a list of thoses comments.
For a better user experience you would like to synchronize this list in realtime
as soon as a new comment is created.
For this matter you will use the EventStreamex.EventListener module in your live view:
defmodule MyApp.CommentLive.Index do
  use MyApp, :live_view

  use EventStreamex.EventListener,
    schema: "comments",
    subscriptions: [%{scopes: [post_id: "posts"]}]

  alias MyApp.Blog
  alias MyApp.Blog.Comment

  @impl true
  def mount(params, session, socket) do
    super(params, session, socket)
  end

  @impl true
  def handle_params(%{"post_id" => post_id} = params, url, socket) do
    {_res, socket} = super(params, url, socket)

    {:noreply,
    socket
    |> stream(:comments, Blog.list_comments(post_id))
    |> assign(:post_id, post_id)
  end

  @impl true
  def handle_info(
        {:on_insert, [{"posts", post_id}], "comments", comment},
        %{assigns: %{post_id: post_id}} = socket
      ) do
    {:noreply,
    socket
    |> stream_insert(:comments, comment.new_record)
  end

  @impl true
  def handle_info(
        {:on_update, [{"posts", post_id}], "comments", comment},
        %{assigns: %{post_id: post_id}} = socket
      ) do
    {:noreply,
    socket
    |> stream_insert(:comments, comment.new_record)
  end

  @impl true
  def handle_info(
        {:on_delete, [{"posts", post_id}], "comments", comment},
        %{assigns: %{post_id: post_id}} = socket
      ) do
    {:noreply,
    socket
    |> stream_delete(:comments, comment.old_record)
  end
end
This code will update the comments list as soon as a new comment is either
inserted, deleted or udpated (notice that here, comments are linked to a post via the post_id field and we load only the comments for a specific post).
For this code to work you will also have to use the EventStreamex.Events module to mark the entity for WAL listening. Without doing so, the comments entity will not be listened from WAL events, and thus, will not be dispatched to operators and live views

  
    
  
  How it works


use EventStreamex.EventListener will do the "magic" by subscribing to the entity
changes in Phoenix.LiveView.mount/3 and Phoenix.LiveView.handle_params/3
callbacks.
That means that if you override these callbacks you have to call the super function
so that the "magic" is done.
The "magic" in question is a subscription to several channels in a pubsub module.
There are 3 kinds of channels this module will automatically subscribe to:
	direct: We subscribe to a specific entity changes (by it's ID)
	unscoped: We subscribes to changes of all entities in a table
	scopes: We subscribe to changes in entities that match a specific scope (like having a specific post_id in the example above)

We can use the 3 kinds a subscriptions at the same time:
use EventStreamex.EventListener,
    schema: "comments",
    subscriptions: [:direct, :unscoped, %{scopes: [post_id: "posts"]}]
By default, we automatically subscribe to the direct and unscoped channels.
All events are received in Phoenix.LiveView.handle_info/2 callbacks with a message of this form:
{:on_insert | :on_delete | :on_update, :direct | [] | [{binary(), id()}], binary(), entity_change()}
More information about each kind of message in the subsections below

  
    
  
  :direct


The :direct channel subscribes to a specific entity by its id field.
This is most usefull in show or edit views where you only need to show one specific entity.
use EventStreamex.EventListener,
    schema: "comments",
    subscriptions: [:direct]
That means that if you want the subscription to be effective, you will have to receive the id of entity
in the params.
Or, pass it manually when you call the super function:
@impl true
def mount(_params, session, socket) do
  super(%{"id" => get_resource_id(session)}, session, socket)
end
The changes in the entity will be received in the Phoenix.LiveView.handle_info/2 callback
with a message of this form:
{:on_insert | :on_update | :on_delete, :direct, binary(), entity_change()}
Here is an example:
defmodule MyApp.CommentLive.Show do
  use MyApp, :live_view

  use EventStreamex.EventListener,
    schema: "comments",
    subscriptions: [:direct]

  alias MyApp.Blog

  @impl true
  def mount(params, session, socket) do
    super(params, session, socket)
  end

  @impl true
  def handle_params(%{"id" => id} = params, url, socket) do
    {_res, socket} = super(params, url, socket)

    {:noreply,
    socket
    |> assign(:id, id)
    |> assign(:comment, Blog.get_comment!(id))}
  end

  @impl true
  def handle_info({:on_insert, :direct, "comments", _comment}, socket) do
    # Should never happen because the entity already exists
    {:noreply, socket}
  end

  @impl true
  def handle_info({:on_update, :direct, "comments", comment}, socket) do
    {:noreply,
    socket
    |> assign(:comment, comment.new_record)
  end

  @impl true
  def handle_info({:on_delete, :direct, "comments", _comment}, socket) do
    # Do some redirection stuff eventually
    {:noreply, socket |> put_flash(:warning, "This comment has been deleted")}
  end
end

  
    
  
  :unscoped


The :unscoped channel subscribes to changes of all entities.
This is what you will use for index views when your entity is not scoped.
use EventStreamex.EventListener,
    schema: "comments",
    subscriptions: [:unscoped]
The changes in the entities will be received in the Phoenix.LiveView.handle_info/2 callback
with a message of this form:
{:on_insert | :on_update | :on_delete, [], binary(), entity_change()}
Here is an example:
defmodule MyApp.CommentLive.Index do
  use MyApp, :live_view

  use EventStreamex.EventListener,
    schema: "comments",
    subscriptions: [:unscoped]

  alias MyApp.Blog
  alias MyApp.Blog.Comment

  @impl true
  def mount(params, session, socket) do
    super(params, session, socket)
  end

  @impl true
  def handle_params(params, url, socket) do
    {_res, socket} = super(params, url, socket)

    {:noreply,
    socket
    |> stream(:comments, Blog.list_comments())
  end

  @impl true
  def handle_info(
        {:on_insert, [], "comments", comment}, socket
      ) do
    {:noreply,
    socket
    |> stream_insert(:comments, comment.new_record)
  end

  @impl true
  def handle_info(
        {:on_update, [], "comments", comment}, socket
      ) do
    {:noreply,
    socket
    |> stream_insert(:comments, comment.new_record)
  end

  @impl true
  def handle_info(
        {:on_delete, [],"comments",  comment}, socket
      ) do
    {:noreply,
    socket
    |> stream_delete(:comments, comment.old_record)
  end
end

  
    
  
  :scopes


The :scopes channel subscribes to changes of all entities that match some
specific field values.
To declare a scoped entity you will do like this:
use EventStreamex.EventListener,
    schema: "comments",
    subscriptions: [%{scopes: [post_id: "posts"]}]
Here, comments have a post_id field related to a posts entity.
You can also have several scopes for an entity (The order matters for the matching in Phoenix.LiveView.handle_info/2):
use EventStreamex.EventListener,
    schema: "comments",
    subscriptions: [%{scopes: [org_id: "organizations", post_id: "posts"]}]
Here, comments have a org_id field related to a organizations entity, and a post_id field, related to a posts entity.
As for :direct scopes, you will have to receive the scope fields in the params so that the module
is able to subscribe to the correct channels.
If your fields are not in the params or are named differently, you will have to pass them yourself to the super function:
@impl true
def handle_params(%{"my_post_id" => my_post_id, "my_org_id" => my_org_id} = params, url, socket) do
  {_res, socket} = super(%{"post_id" => my_post_id, "org_id" => my_org_id}, url, socket)

  {:noreply,
  socket
  |> stream(:comments, Blog.list_comments(my_post_id, my_org_id))
end
Events will be received in the c:Phoenix.LiveView.handle_info/2 callback, with messages of this form:
{:on_insert | :on_update | :on_delete, [{"related_scoped_entity", scope_id}], binary(), entity_change()}
For instance, with our previous example, an insert event message will look like this:
{:on_insert, [{"organizations", org_id}, {"posts", post_id}], "comments", entity_change}
The order of scopes will be the same as the one you specified above in the use.
Here is a full example:
defmodule MyApp.CommentLive.Index do
  use MyApp, :live_view

  use EventStreamex.EventListener,
    schema: "comments",
    subscriptions: [%{scopes: [post_id: "posts"]}]

  alias MyApp.Blog
  alias MyApp.Blog.Comment

  @impl true
  def mount(params, session, socket) do
    super(params, session, socket)
  end

  @impl true
  def handle_params(%{"post_id" => post_id} = params, url, socket) do
    {_res, socket} = super(params, url, socket)

    {:noreply,
    socket
    |> stream(:comments, Blog.list_comments(post_id))
    |> assign(:post_id, post_id)
  end

  @impl true
  def handle_info(
        {:on_insert, [{"posts", post_id}], "comments", comment},
        %{assigns: %{post_id: post_id}} = socket
      ) do
    {:noreply,
    socket
    |> stream_insert(:comments, comment.new_record)
  end

  @impl true
  def handle_info(
        {:on_update, [{"posts", post_id}], "comments", comment},
        %{assigns: %{post_id: post_id}} = socket
      ) do
    {:noreply,
    socket
    |> stream_insert(:comments, comment.new_record)
  end

  @impl true
  def handle_info(
        {:on_delete, [{"posts", post_id}], "comments", comment},
        %{assigns: %{post_id: post_id}} = socket
      ) do
    {:noreply,
    socket
    |> stream_delete(:comments, comment.old_record)
  end
end

  
    
  
  Entity change structure


The event received in the Phoenix.LiveView.handle_info/2 callbacks have information about
the entity and its changes.
Here is what the structure looks like:
%WalEx.Event{
  name: atom(),
  type: :insert | :update | :delete,
  source: %WalEx.Event.Source{
    name: String.t(),
    version: String.t(),
    db: String.t(),
    schema: String.t(),
    table: String.t(),
    columns: map()
  },
  new_record: map() | nil,
  old_record: map() | nil,
  changes: map() | nil,
  timestamp: DateTime.t(),
  lsn: {integer(), integer()}
}
I am using the WalEx package internally
	name: The name of the entity (ie: :comments for a table named comments)
	type: The type of event between insert, update and delete
	source: Information about the event:	name: "WalEx"
	version: Current version of WalEx
	db: The name of the database
	schema: Mostly "public"
	table: The name of the table (ie: "comments")
	columns: A map of fields with their type (ie: %{"id": "integer", "message": "varchar"})


	new_record: The entity itself for insert and update events. nil for delete events.
	old_record: The entity itself for delete events. nil for insert and update events.
	changes: A map with the changes in the entity in update events, nil otherwise (see below)
	timestamp: The timstamp of the event in DateTime type
	lsn: A tuple containing information about the publication cursor


  
    
  
  changes


When you receive an update event, you will also have the changes field set to a map containing
the changes the entity received since the update.
This map contains the changed fields as keys, and a map describing the change as value.
This "change" map contains 2 fields:
	old_value: The value before the update
	new_value: The value after the update

For instance, let's say you have a comments entity with 4 fields: id, message, rating, timestamp.
You have a comment with these values:
%Comment{
  id: "dd4bc2ba-c7cc-4a05-a1c7-9f26cd9ab79f",
  message: "This is my first comment",
  rating: 4,
  timestamp: "2024-07-23T12:00:00Z"
}
Now, the comment is update this these new values:
%Comment{
  id: "dd4bc2ba-c7cc-4a05-a1c7-9f26cd9ab79f",
  message: "This is (not) my first comment (anymore)",
  rating: 5,
  timestamp: "2024-07-23T12:00:00Z"
}
The event structure will look like this:
%WalEx.Event{
  name: :comments,
  type: :update,
  source: %WalEx.Event.Source{
    name: "WalEx",
    version: "4.4.0",
    db: "postgresql",
    schema: "public",
    table: "comments",
    columns: %{
      id: "uuid",
      message: "varchar",
      rating: "integer",
      timestamp: "datetime"
    }
  },
  new_record: %Comment{
    id: "dd4bc2ba-c7cc-4a05-a1c7-9f26cd9ab79f",
    message: "This is (not) my first comment (anymore)",
    rating: 5,
    timestamp: "2024-07-23T12:00:00Z"
  },
  old_record: nil,
  changes: %{
    message: %{
      old_value: "This is my first comment",
      new_value: "This is (not) my first comment (anymore)"
    },
    rating: %{
      old_value: 4,
      new_value: 5
    }
  },
  timestamp: "2024-08-25T13:13:30Z",
  lsn: {0, 0}
}

  
    
  
  Unsubscribing from events


The unsubscribe from events is done automatically in when the process exists (each LiveView is monitored).
You do not have anything to do.

  
    
  
  Handling subscriptions later


If you need, for any reason, to handle subscriptions at another moment than the mount and handle_params callbacks,
we provide the handle_subscriptions/2 function.
This can be useful if the parameters used for scoped channels
are handled in a different way than just getting them from query parameters.

  
    
  
  Params


	socket: The current socket
	params: A map containing the parameters to use for scoped channels. Each scope field must be present in the params map as a string


  
    
  
  Returns


The updated socket

  
    
  
  Example


defmodule MyModule do
  use EventStreamex.EventListener,
    schema: "comments",
    subscriptions: [%{scopes: [post_id: "posts"]}]

  def handle_params(%{"id" => id}, _url, socket) do
    entity = MyEntity.get(id)

    {:noreply,
      handle_subscriptions(
        socket,
        %{"post_id" => entity.post_id}
      )}
  end
end
In this example, I need a scoped channel with a post_id field.
This field is not present in the url (which is a bad practice), but
I know I can find it in the entity I fetch.
Here, I don't have to call the super() function because I do not need
the EventListener to handle the subscriptions as I do it manually.
You don't have to handle the unsubscribe either because it will be done for you
when the LiveView exits.

  
    
  
  Subscribing to other entities' events


You have 2 ways to listen for several entities

  
    
  
  Dynamically


Let's say you display posts and handle their creation.
And when a new post is created you want to redirect to the detail page
of this new post.
But, the enitty used for the detail of this post is a derived entity from
an operator.
Thus, you can't be sure the entity will exist when you redirect to its detail page
because of this asynchronicity of the event streaming architecture.
So you would like to listen for this entity beeing created.
Hopefully, you know its ID because you just created the post and this derived
entity uses the same ID.
So you can call the subscribe_entiy/4 function to listen for the events coming from
this derived entity.
subscribe_entity(socket, "post_with_comments_count", :direct, %{"id" => post_id})
Params
	socket: The socket
	entity_name: The entity name to listen to as a string
	subscription: The kind of channel you want to listen to (these are the same as for the module configuration: :direct, :unscoped, %{scopes: []})
	params: A map with the parameters needed for :direct and :scopes scopes.

Return value
The updated socket
Received events
Events are received the same way as other events but related to the entity:
@impl true
def handle_info({:on_insert, :direct, "post_with_comments_count", post}, socket) do
  {:noreply, socket |> push_navigate(to: "/posts/#{post.id}")}
end

  
    
  
  From the using macro


You can also subscribe to several entities directly when you use the EventListener.
Insead of using schema and subscriptions, you can use schemas with an array of subscriptions.
This array can be:
	An array of schemas (binaries)
	An array of structures with schema and subscriptions (with subscriptions being optional)

Some examples:
# Listens to `comments` with a scope
use EventStreamex.EventListener,
    schema: "comments",
    subscriptions: [%{scopes: [post_id: "posts"]}]

# Listens to `comments` and `posts` with default `direct` and `unscoped` scopes
use EventStreamex.EventListener, schemas: ["comments", "posts"]

# Same
use EventStreamex.EventListener, schemas: [%{schema: "comments"}, %{schema: "posts"}]

# Now, they both listen only for `direct` events
use EventStreamex.EventListener, schemas: [%{schema: "comments"}, %{schema: "posts"}], subscriptions: [:direct]

# Now, while `comments` listens for `unscoped`, `posts` still listens for `direct` events
use EventStreamex.EventListener, schemas: [%{schema: "comments", subscriptions: [:unscoped]}, %{schema: "posts"}], subscriptions: [:direct]

  
    
  
  Unsubscribing to other entities' events


Use the function unsubscribe_entity/3 to manually unsubscribe from an event's events.
unsubscribe_entity(socket, "post_with_comments_count", :direct)
The params will be the same used for subscription so no need to pass them again.

  
    
  
  Params


	socket: The socket
	entity_name: The name of the entity to stop linstening
	subscription: The kind of channel to stop listening to (these are the same as for the module configuration: :direct, :unscoped, %{scopes: []})


  
    
  
  Return value


The updated socket

  
    
  
  use params


When you use this module, you will be able to specify these parameters:
	:schema: The name of the entity as a string (mandatory field)
	:subscriptions: The subscriptions you want to do (Defaults to [:direct, :unscoped])
	:application: The application module to use to retrieve config values (Defaults to Application)

EventStreamex.Events 
    



      
This module can be used in schema modules and is necessary to mark an
entity for event listening.
This is what connects an entity to the WAL events.
If you don't use this module for an entity you won't be able to receive
its events, which will prevent you from listening to events in live views (EventStreamex.EventListener),
and to use them in operators to update other entities (EventStreamex.Operators.Operator).
To mark an entity for event listening, you only have to use this module in
the schema module of the entity:
defmodule MyApp.Blog.Comment do
  use MyApp.Schema

  use EventStreamex.Events,
    name: :my_app,
    schema: "comments",
    extra_channels: [%{scopes: [post_id: "posts"]}]

  import Ecto.Changeset

  schema "comments" do
    field :message, :string
    field :post_id, Ecto.UUID

    timestamps(type: :utc_datetime)
  end

  @doc false
  def changeset(comment, attrs) do
    comment
    |> cast(attrs, [:id, :message, :post_id])
    |> validate_required([:message, :post_id])
  end
end
In this example, we have a comments entity that we want to mark for event listening.
This comments entity is scoped by a post_id related to a posts entity.
So we needed to add an extra_channels so that comment events will also be emitted in a channel
specific for his scopes (See EventStreamex.EventListener for more details on scopes)
Several scopes can be used for an entity. The order used for scopes must be the same
when you define it in EventStreamex.EventListener.
Here is an example with another scope:
use EventStreamex.Events,
    name: :my_app,
    schema: "comments",
    extra_channels: [%{scopes: [org_id: "organizations", post_id: "posts"]}]
Here we say a comment has 2 scopes: an organization and a post.
That means the entity must have the 2 fields: org_id and post_id.

  
    
  
  How it works


As said earlier, this module marks the entity for WAL events.
Which allows you to listen to its changes and to use it in operators.
Under the hoods, this module will create a sub module inside the module you use it in.
And it will define functions that will be called each time something happens with this entity (insert, update, delete).
Then, it will dispatch the event to all operators listening to this entity (EventStreamex.Operators.Operator).
And it will emit an event in the configured pubsub module in several channels:
	direct: We emit in a specific channel containing the ID of the enity
	unscoped: We emit in a channel for changes of all entities of this type
	scopes: We emit in a channel for entities matching the given scopes (Like in the example above)

For more information about scopes and channels, see EventStreamex.EventListener.

  
    
  
  use params


When you use this module, you will be able to specify these parameters:
	:module_name: The name of the module that will be created in your module to perform the events logic (Defaults to "Events")
	:schema: The name of the entity as a string (mandatory field)
	:name: The name of your application (Mandatory field)
	:extra_channels: The channels to emit events to. :direct and :unscoped are always used, and you can add your own scopes for this entity (Defaults to [])
	:application: The application module to use to retrieve config values (Defaults to Application)
	:scheduler: If you want to use another scheduler to handle events processed by operators (Defaults to EventStreamex.Operators.Scheduler)

EventStreamex.Events.LiveViewMonitor 
    



      
This GenServer will monitor every LiveView of the current node when it uses EventListener.
A LiveViewMonitor is started in each of your nodes.
When a LiveView exits, it will call EventStreamex.EventListener.unsubscribe_all/2 which
will unsubscribe from all channels.

EventStreamex.Operators.EntityProcessed 
    



      
This module is called last for each event.
All it does is register the last updated timestamp of the entity and store it
in the EventStreamex.Operators.ProcessStatus module.
By default the timestamp is taken from the updated_at field of the entity
but you can change that in the :process_status_entity_field config param.
If no timestamp could be found, the entity is not registered in the process status.
Works with fields of type:
	bigint: The unix timestamp with a microsecond precision
	DateTime: Using microsecond precision
	NaiveDateTime: Using microsecond precision

Executes the operator tasks when an event is received.
A new executor is started by the EventStreamex.Operators.Scheduler as soon
as an event is received.
It is responsible to start the operator the scheduler gave it to handle the
new event.
And it will handle the operator's failures.
Meaning that if the operator crashes or returns an error, the executor will
restart until the operator successfully finishes its task or has failed to many times.

  
    
  
  Restart strategy


In the configuration you can set the restart strategy:
config :event_streamex,
  operator_queue_backoff_multiplicator: 2,
  operator_queue_max_restart_time: 10000,
  operator_queue_max_retries: 5,
  operator_queue_min_restart_time: 500
	operator_queue_backoff_multiplicator: At each new retry, the last wait time is multiplied by this multiplicator to get the new time (ie: If it waited 1.5 second for the last retry and the multiplicator is 2, then, it will now wait 3 seconds before the next restart).
	operator_queue_max_restart_time: The time we wait before a restart is maxed at this value.
	operator_queue_max_retries: The maximum number of times we allow the operators to retry after a failure.
	operator_queue_min_restart_time: The minimum time we will wait before a restart.

At each failure of an operator, the error logger is called (EventStreamex.Operators.Logger.ErrorLoggerAdapter)
as well as a telemetry.
When an operator exceeds the maximum number of restarts the executor crashes as
well as the EventStreamex.Operators.Scheduler and no more event will be processed until you
restart the scheduler with EventStreamex.restart_scheduler/0.
This is because the order of the events is very important and we cannot allow an
event to fail, otherwise it would compromise your data.
You can create your own EventStreamex.Operators.Logger.ErrorLoggerAdapter and add it to the configuration:
config :event_streamex,
  error_logger_adapter: {EventStreamex.Operators.Logger.LoggerAdapter, []}
Doing so, you could implement some kind of notification system when there is a
failure to warn your team about it.
When a failure happens, maybe it's because of a temporary failure and you could just restart the scheduler.
But maybe the failure is due to a coding failure.
In this case you should correct the issue, ship a new version of your app and restart the scheduler (or the entire application).
With a database queue adapter (the default one), the queue is saved to database and will be loaded again when the
application restarts.
You won't miss any event.
The executor is used internally, you should never call it directly

Behaviour to handle errors with operators (EventStreamex.Operators.Operator).
When implementing this behaviour, you must put use a GenServer.
There is currently one adapter available and used by default: EventStreamex.Operators.Logger.LoggerAdapter.
This adapter simply logs the error to the console.
The executor (EventStreamex.Operators.Executor) who catches the operator failure and
calls the error logger will also execute a telemetry event so that you can
see errors over time in a chart.
There are 2 kinds of errors:
	Retry: The operator failed but will retry soon
	Failed: The operator retried several times but failed again.

For more information about the retry strategy, see EventStreamex.Operators.Executor.

An error logger adapter that writes logs in the console.
See EventStreamex.Operators.Logger.ErrorLoggerAdapter for more information
about the error logging.

An operator listens for events and will usually create/update/delete
another entity based on those events.
This is a very usefull pattern. Here is why:

  
    
  
  The issue without event streaming


Let's say you have a posts entity which contains:
	title: a title
	body: a body

You also have a comments entity which contains:
	post_id: The ID of the post it is attached to
	comment: The comment content

When you show a list of your posts you might want to show the number
of comments it contains.
To do so you will be tempted to make a SQL query with a subquery that will
count the comments of your posts.
This is pretty trivial:
SELECT
  p.id AS id,
  p.title AS title,
  p.body AS body,
  (SELECT COUNT(*) FROM comments c WHERE c.post_id = p.id) AS comments_count
FROM posts p;
But it has several flaws.
Let's see them in more details.

  
    
  
  Your queries will become slower and slower over time


Because subqueries are not free. They take more time to execute that a simple
query with no join and subquery.
So, the more posts and comments you have, the slower the query will be.

  
    
  
  You will end up adding complexity in your queries


As your app evolves, you will probably create new entities related to your posts and comments.
Thus, you will add more and more subqueries and joins to your queries, making them slower and harder to maintain...

  
    
  
  Realtime will become a mess to maintain


If you want to handle realtime events like updating the number of comments when a new one
is created it will be a mess to maintain overtime because you will end up subscribing to a lot of events
as you link your posts to other entities.
And you will probably forget to subscribe to some of the events making your app
inconsistant (some features will be in realtime and other won't be).

  
    
  
  Almost impossible to maintain recursivity


If your comments can have subcomments it will be almost impossible
to maintain the comments count because of the recursivity it implies.

  
    
  
  Depending on entities not directly related to the base entity


Let's say you create another entity named ratings which saves each rating given
to a specific comment.
It would look like this:
	comment_id: The ID of the comment related to this rating
	rating: The given rating

Now imagine you want to display for every comment, the average rating and
the total count of ratings for this comment.
Here again, you could solve this with some subqueries.
But what if you want to show the average rating of all comments in the post
so that users can see that a post has received a lot of useful comments.
I don't even want to think about the query to build to do that...

  
    
  
  Benefits from using event streaming


Let's come back to our posts and comments.
Now let's imagine that instead of our subquery to count the comments we use
an event streaming architecture and create another entity whose goal is
specifically to show the posts with their comments count.
This post_with_comments_counts entity will look like that:
	title: a title
	body: a body
	comments_count: the number of comments of the post

And this entity will listen to new comments to update its comments count:
when a new comment is created it will increase the number of comments of the related post by 1.
With EventStreamex it would look like that:
defmodule MyApp.Operators.PostWithCommentsCountOperator do
  use EventStreamex.Operators.Operator,
    schema: "post_with_comments_counts",
    repo: MyApp.Repo

  on_event(
    all_events(),
    entities(["comments", "posts"]),
    query(
      "MERGE INTO post_with_comments_counts dest
      USING (
          SELECT
              p.id AS id,
              p.title AS title,
              p.body AS body,
              p.inserted_at AS inserted_at,
              p.updated_at AS updated_at,
              (SELECT COUNT(c.id) FROM comments c WHERE c.post_id = p.id) AS comments_count
          FROM posts p
          WHERE p.id = $1
      ) src ON src.id = dest.id
      WHEN NOT MATCHED THEN
        INSERT (id, title, body, inserted_at, updated_at, comments_count)
        VALUES (src.id, src.title, src.body, src.inserted_at, src.updated_at, src.comments_count)
      WHEN MATCHED THEN
        UPDATE SET
            title = src.title,
            body = src.body,
            inserted_at = src.inserted_at,
            updated_at = src.updated_at,
            comments_count = src.comments_count;",
      id([{"comments", :post_id}, {"posts", :id}]),
      delete_with(["posts"])
    )
  )
end
Here, we use the EventStreamex.Operators.Operator module to listen for
posts and comments in all kinds of events (insert, update, delete).
And each time a post or a comment is inserted, updated, deleted, we perform this MERGE query
to create/update our post_with_comments_counts.
The MERGE query will insert a new post if the post does not exist and update
it if it already exists.
id/1 is used to map the fields of the listened entities to the id of the
post_with_comments_counts entity (id for a post and post_id for a comment).
Finally, delete_with/1 tells the operator which entities deletion should
also delete the related post_with_comments_counts entity.
As you can see, this MERGE query also uses a subquery, so why is it better
than our first implementation?

  
    
  
  This is a naive implementation


The query/3 macro is provided for simplicity.
But you can also replace it by a function to do things as you think it's best.
This function will receive the event, allowing you to do whatever you want with it.

  
    
  
  The query is executed in the background


In our first example, the query was executed everytime someone wants to
display the list of posts.
In our event streaming example, the query is executed in the background
each time there is an event with a post or a comment.
To display your list of posts you will just have to do:
SELECT * FROM post_with_comments_counts;
Which is simpler and faster than our former query.

  
    
  
  Realtime is easy


You can use the module EventStreamex.EventListener to easily add realtime
in your live view applications.
Because we are using PostgreSQL WAL events, everytime something happens on an entity
(post_with_comments_counts as well), an event is sent.
This event will be processed by operators listening to it (like our operator here does for posts and comments),
and it will be sent to a pubsub processor.
The EventStreamex.EventListener, now just has to listen to this pubsub processor
to notify your live view the entity your are showing as been updated.
Everything, now, can be realtime, without beeing hard to maintain.

  
    
  
  A better pattern to model your data


With this approach you keep your entities consistent and simple.
And you create specialized entities for what you want to show in your views.
A post will stay a post with its title and body fields and a comment will
stay a comment with a comment field.
post_with_comments_count is just a specialized entity used for your view.
This keeps your data and APIs consistent and simple with explicit fields and names.
For instance, let's say you have a users entity with a login and a password field.
But later on, you realize that you need to store some information about your users,
like its name, its address, and so on.
You will be tempted to update your users entity to add these new fields:
	login
	password
	address
	name

But still, in your application, when comes the moment to create your user,
you only need the login and password.
So what do you do? You set the other fields as optionals in your API and set them
a default value (empty string or null).
And in the view that allows a user to specify more information about him (name and address),
you will either:
	Use the same API endpoint as for the users creation
	Or create a new dedicated API endpoint

But both approachs are bad practices!
In the first approach that means that your API expects 2 mandatory fields: login and password
and 2 optional fields: name and address.
But here, there is no sense in sending the login and password.
And maybe you would like the name and address fields to be mandatory.
But because your API set these fields as optionals, you will have to check for them from the client side.
Which is a bad practice.
In the second approach, that means that you will create an API endpoint for the user profile.
But this API, will finally update the users entity.
You end up with hard to understand APIs because their name does not match
to the entity being modified.
A better approach would be to have a users entity with:
	login
	password
And its own API endpoint /users with only these two fields.

And you would have a user_profiles entity with:
	user_id
	name
	address
And its own API endpoint /users/{user_id}/user_profiles with only the name and address fields.

And what if you need to display all of these information in your view?
Well, you use an operator like we did for the post_with_comments_counts.
Named, for instance: user_with_profiles, and listening to users and user_profiles events.
This way, you keep your data and API consistent.

  
    
  
  use params


When you use this module, you will be able to specify these parameters:
	:schema: The name of the entity as a string (mandatory field)
	:repo: The repository used for the connection to the database (mandatory field)
	:sql_adapter: The SQL adapter to use (Defaults to Ecto.Adapters.SQL)
	:id: The field used as an ID for the schema (Defaults to :id)

Handle the process status of entities.
Everytime an event is processed, it's saved in the ProcessStatus along
with the processed timestamp.
Thus, the process status contains the latest processed timestamp for all entities.
You can use this module to ensure an entity has been processed.
Let's say you have posts and post_with_comments_counts entities.
When a posts is created, an operator, listening for posts will create
a post_with_comments_counts entity with the same ID.
You have an API endpoint to create a new posts (POST /posts), and as soon as your appliation
calls it, it redirects to another view requesting the detail of the created post_with_comments_counts (GET /post_with_comments_counts).
But the creation of the later is asynchronous, so you may receive a 404 error from your API.
In order to avoid that, you could implement a mecanism that would return, along
with the posts newly created, the post creation timestamp.
Then, in the GET /post_with_comments_counts endpoint, you would send along the timestamp you got
from the POST /posts endpoint.
And in your controller you would check if the entity has been
processed before returning the post_with_comments_counts requested.
You can use processed?/2 in order to do so.
For instance, when you create a posts, you receive this header from the API:
X-Event-Streamex-Entity-Timestamp: "posts/1422057007123000"
(You can use the :process_status_entity_field config parameter to set
the field to use in your entities to know the last updated time.
You should return the field's value in your API, and use this value
in the processed?/2 function)
You will send this header when you call the GET /post_with_comments_counts endpoint.
And the controller will immediately check if the entity has been processed:
EventStreamex.Operators.ProcessStatus.processed?("posts", 1422057007123000)
If the entity has not been processed yet, then, you know the post_with_comments_counts entity
has not been created yet (it works the same for updated entities).
So you can immediately return a 425 response for instance, informing the client to retry later.
If the entity has been processed, then you fetch it and return it.
If you get a 404 error, this time, that means the entity did not exist.

A database queue adapter.
The queue is stored in the database.
It uses the information from the configuration to connect to the database
and it automatically creates the database to store the queue.

Do not store the process status.
This adapter can be used to improve performances.
This adapter should only be used for tests or dev environments as it will
not recover the process status in case of a node crash in a distributed system.

Behaviour for the process status storage.
Everytime an event is received we will save its name and timestamp to
keep a process status up to date.
The process status allows us to know for each entity the last time it has been processed.
Here, "processed" means all of these steps:
	Event received from WAL
	Event emitted in pubsub
	Event processed by all operators listening to it

The adapters keeps a copy of this process status, mostly in case of crash (Except for EventStreamex.Operators.ProcessStatus.NoAdapter which does not saves the process status and should not be used in production)
There are currently 2 process status adapters:
	EventStreamex.Operators.ProcessStatus.DbAdapter: A process status that uses the database tu store its items (This is the default process status adapter)
	EventStreamex.Operators.ProcessStatus.NoAdapter: A process status that do not store its items

You can create your own adapter and set the config to use it.
Here is the full code of a possible memory adapter:
defmodule EventStreamex.Operators.ProcessStatus.MemAdapter do
  use GenServer
  @behaviour EventStreamex.Operators.ProcessStatus.ProcessStatusStorageAdapter

  @impl true
  def start_link(opts) do
    GenServer.start_link(__MODULE__, opts, name: __MODULE__)
  end

  @impl EventStreamex.Operators.ProcessStatus.ProcessStatusStorageAdapter
  def item_processed(item) do
    GenServer.call(__MODULE__, {:save, item})
  end

  @impl EventStreamex.Operators.ProcessStatus.ProcessStatusStorageAdapter
  def load() do
    GenServer.call(__MODULE__, :load)
  end

  @impl EventStreamex.Operators.ProcessStatus.ProcessStatusStorageAdapter
  def reset() do
    GenServer.call(__MODULE__, :reset)
  end

  # Callbacks

  @impl true
  def init(_opts) do
    {:ok, %{}}
  end

  @impl true
  def handle_call({:save, {entity, timestamp}}, _from, process_status) do
    new_process_status = process_status |> Map.put(entity, timestamp)
    {:reply, {:ok, new_process_status}, new_process_status}
  end

  @impl true
  def handle_call(:load, _from, process_status) do
    {:reply, {:ok, process_status |> Map.to_list()}, process_status}
  end

  @impl true
  def handle_call(:reset, _from, _process_status) do
    {:reply, {:ok, []}, []}
  end
end
The adapter must be a GenServer and declare the function start_link/1.
Then, it must provide these functions:
	item_processed/1: Updates the timestamps of an entity (or adds it if the entity was not known before)
	load/0: Return the entire process status
	reset/0: Remove all items from the process status

An item is a tuple containing:
	The name of the entity (as a string)
	The timestamp of the last event processed for this entity

Handle the queue of operators to process.
Every event received is dispatched to operators (EventStreamex.Operators.Operator),
and executed sequentially to keep ordenancing.
Every time an event is received, a task is added to the queue of events with each
operator listening for this event.
And everytime a an operator task is finished, it is update in the queue, until it is deleted completely when all operators have finished their task.
A queue item is a tuple containing the operator modules with their completion status and the event to process.
The queue is handled internally.
You should not use it directly.

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    EventStreamex.Operators.Queue.DbAdapter - EventStreamex v1.4.3
    
    

    


  
  

    
EventStreamex.Operators.Queue.DbAdapter 
    



      
A database queue adapter.
The queue is stored in the database.
It uses the information from the configuration to connect to the database
and it automatically creates the database to store the queue.

Adapter that does nothing with the queue.
This adapter simply improves performances.
This adapter should only be used for tests or dev environments as it will
not recover the queue in case of a node crash in a distributed system.

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    EventStreamex.Operators.Queue.QueueStorageAdapter - EventStreamex v1.4.3
    
    

    


  
  

    
EventStreamex.Operators.Queue.QueueStorageAdapter behaviour
    



      
Behaviour for the queue storage.
Every event received is dispatched to operators (EventStreamex.Operators.Operator),
and executed sequentially to keep ordenancing.
Every time an event is received, a task is added to the queue of events with each
operator listening for this event.
And everytime a an operator task is finished, its completion status is updated in the task.
When all operators have finished their task for this event, it is removed from the queue.
The adapters keeps a copy of this queue, mostly in case of crash (Except for EventStreamex.Operators.Queue.NoAdapter which does not save the queue and should not be used in production)
There are currently 2 queue adapters:
	EventStreamex.Operators.Queue.DbAdapter: A queue that uses the database tu store its items (This is the default queue adapter)
	EventStreamex.Operators.Queue.NoAdapter: A queue that does not store its items

You can create your own adapter and set the config to use it.
Here is the full code of a possible memory adapter:
defmodule EventStreamex.Operators.Queue.MemAdapter do
  use GenServer
  @behaviour EventStreamex.Operators.Queue.QueueStorageAdapter

  @impl true
  def start_link(opts) do
    GenServer.start_link(__MODULE__, opts, name: __MODULE__)
  end

  @impl EventStreamex.Operators.Queue.QueueStorageAdapter
  def add_item(item) do
    GenServer.call(__MODULE__, {:save, item})
  end

  @impl EventStreamex.Operators.Queue.QueueStorageAdapter
  def delete_item(item) do
    GenServer.call(__MODULE__, {:delete, item})
  end

  @impl EventStreamex.Operators.Queue.QueueStorageAdapter
  def update_processors_status(item) do
    GenServer.call(__MODULE__, {:update, item})
  end

  @impl EventStreamex.Operators.Queue.QueueStorageAdapter
  def load_queue() do
    GenServer.call(__MODULE__, :load)
  end

  @impl EventStreamex.Operators.Queue.QueueStorageAdapter
  def reset_queue() do
    GenServer.call(__MODULE__, :reset)
  end

  # Callbacks

  @impl true
  def init(_opts) do
    {:ok, []}
  end

  @impl true
  def handle_call({:save, item}, _from, queue) do
    new_queue = queue ++ [item]
    {:reply, {:ok, new_queue}, new_queue}
  end

  @impl true
  def handle_call({:delete, {id, _item}}, _from, queue) do
    new_queue = queue |> Enum.reject(fn {i, _t} -> i == id end)
    {:reply, {:ok, new_queue}, new_queue}
  end

  @impl true
  def handle_call({:update, {id, item}}, _from, queue) do
    new_queue = queue |> Enum.map(fn {i, t} ->
      if i == id do
        item
      else
        t
      end
    end)
    {:reply, {:ok, new_queue}, new_queue}
  end

  @impl true
  def handle_call({_action, nil}, _from, queue) do
    {:reply, {:ok, queue}, queue}
  end

  @impl true
  def handle_call(:load, _from, queue) do
    {:reply, {:ok, queue}, queue}
  end

  @impl true
  def handle_call(:reset, _from, _queue) do
    {:reply, {:ok, []}, []}
  end
end
The adapter must be a GenServer and declare the function start_link/1.
Then, it must provide these functions:
	add_item/1: Add a new item to the queue
	delete_item/1: Delete an item from the queue
	update_processor